

LOGISTIC NUMBERS**PURPOSE**

Generate a sequence of logistic numbers.

DESCRIPTION

The logistic sequence is defined by the following equation:

$$X_{n+1} = k * X_n * (1 - X_n)$$

where k is a user defined constant. The user also provides a starting value (i.e., X_1).

SYNTAX

LET <var> = LOGISTIC NUMBERS FOR I = <start> <inc> <stop>

where <var> is a variable where the logistic numbers are stored;

<start> is a number or parameter that is the first element of <var> in which the logistic numbers are stored (it is almost always 1);

<inc> is a number or parameter that specifies the row increment in <var> for storing the logistic numbers (it is almost always 1);

and <stop> is a number or parameter that specifies the last row of <var> in which to store the logistic numbers.

EXAMPLES

LET YLOG = LOGISTIC NUMBERS FOR I = 1 1 100

NOTE 1

The user must specify the constant and the starting value as follows:

LET X0 = <value>

LET K = <value>

where X0 should be a decimal number between 0 and 1 (inclusive) and K should be a decimal number between 0 and 4 (exclusive).

NOTE 2

This sequence converges to 0 for values of K less than 1.0. For values of K between 1 and 3 it converges to a single number. For values of K greater than 3 it begins to exhibit chaotic behavior.

DEFAULT

None

SYNONYMS

None

RELATED COMMANDS

FRACTAL PLOT	=	Generate a fractal plot.
CANTOR NUMBERS	=	Generate a sequence of Cantor numbers.
FRACTAL	=	Generate a fractal sequence.
PHASE PLANE DIAGRAM	=	Generate a phase plane diagram.
PLOT	=	Plots data or functions.

REFERENCE

“Chaos, Fractals, and Dynamics,” Robert Devine, Addison-Wesley, 1990, (pp. 26-31).


APPLICATIONS

Chaos

IMPLEMENTATION DATE

88.7

```
PROGRAM 1
  DIMENSION 20 COLUMNS
  LET KINC=0.3
  LET X0 = 0.05
  LOOP FOR L = 1 1 10
    LET K = (L-1)*KINC + 0.5
    LET Y^L = LOGISTIC NUMBERS FOR I = 1 1 40
  END OF LOOP
  LET X = SEQUENCE 1 1 40
  XLIMITS 0 40
  XTIC OFFSET 2 2
  YLIMITS 0 0.8
  YTIC OFFSET 0.1 0
  TITLE LOGISTIC SEQUENCE FOR 10 VALUES OF K
  LINE SOLID DA DA DA DA DA DA DA DA DA DO
  PLOT Y1 Y2 Y3 Y4 Y5 Y6 Y7 Y8 Y9 Y10 VS X
```



PROGRAM 2

```

ERASE; PRE-ERASE OFF
TITLE BIFURCATION PLOT FOR LOGISTIC SEQUENCE
X1LABEL K PARAMETER; Y1LABEL ASYMPTOTIC VALUES OF SEQUENCE
XLIMITS 0 4; XTIC OFFSET 0.1 0.1; YLIMITS 0 1; YTIC OFFSET 0.1 0.1
CHARACTER .; LINE BLANK; CHARACTER JUSTIFICATION LEBO; CHARACTER SIZE 0.1
. DO INTERVAL FOR K BETWEEN 0 AND 1
LET X0 = 0.05; LET NMAX = 200
LET NCUT = 150; LET TAG = SEQUENCE 1 1 NMAX
LOOP FOR K = 0 0.1 1.0
    LET Y = LOGISTIC NUMBERS FOR I = 1 1 NMAX; RETAIN Y SUBSET TAG > NCUT
    LET YD = DISTINCT Y; LET N = SIZE YD
    LET X = K FOR I = 1 1 N
    PLOT YD VS X
    DELETE YD X; TITLE; FRAME OFF; X1LABEL; Y1LABEL
END OF LOOP
. DO INTERVAL FOR K BETWEEN 1 AND 3
LOOP FOR K = 1.05 0.05 3.0
    LET Y = LOGISTIC NUMBERS FOR I = 1 1 NMAX; RETAIN Y SUBSET TAG > NCUT
    LET YD = DISTINCT Y; LET N = SIZE YD
    LET X = K FOR I = 1 1 N
    PLOT YD VS X; DELETE YD X
END OF LOOP
. DO INTERVAL FOR K BETWEEN 3 AND 4
LOOP FOR K = 3.01 .01 4.0
    LET Y = LOGISTIC NUMBERS FOR I = 1 1 NMAX; RETAIN Y SUBSET TAG > NCUT
    LET YD = DISTINCT Y; LET N = SIZE YD
    LET X = K FOR I = 1 1 N
    PLOT YD VS X; DELETE YD X
END OF LOOP

```

